State_Variable_f_1Hz_10K

```
 *   dsauersanjose@aol.com
 *   www.idea2ic.com                                           R3 = 1/A0
 *   4/15/08                     ___________________________/\  /\  /\___
 *                              |                             \/  \/     |
 *                              |                  R2 = 1/A1             |
 *                              |______________/\  /\  /\_               |
 *                              |                \/  \/   |              |
 *                              |    R1 = 1               |BP            |LP
 *                              |_/\  /\  /\_ HP          |              |
 *             R0 = 1/A0        |   \/  \/   |            |              |
 *       ___                    ||\          |  XONE_S1   |  XONE_S2     |
 *      |IN |_/\  /\  /\__|_|-\          |  _______   |  _______     |  ___
 *      |___|   \/  \/  SUMN|   \________|_|       |_|__|       |__|_|OUT|
 *        _|_               |   /          |  1/s  |    |  1/s  |      |___|
 *       /VIN\            __|+/ OPA1       |_______|    |_______|
 *       \___/           |  |/
 *         |            _|_          H(s) = 1/(s^2 +s*A1   + A0)
 *         |            ///          H(s) = 1/(s^2 +s*w0/Q + w0^2)
 *        ///
 *            Set  A0 = 1 and scale s to 1KHz
 *            Then R2 = Q and  s = 2*PI*1KHz

.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN     0      AC     1      DC     0
R0        VIN     SUMN   10k
R1        SUMN    HP     10k
R2        SUMN    BP     100k
R3        SUMN    LP     10k
XOPA1     SUMN    0      HP     OPA
XONES1    HP      BP            ONE_S
XONES2    BP      LP            ONE_S
.ac       dec     50     10     10k

*==The_Absolute_Value_Of_the_Resistors_Does_Not_Matter====
.control
run
plot  db(bp)  db(hp) db(lp)  title  StateVariable_Q_10
*====Only_The_Ratios_Are_Important================
alter  R2     resistance =   10k
run
plot db(bp)  db(hp) db(lp)   title  StateVariable_Q_1
.endc
*
*
*              C1 .0159uF
*              _||______ O1     _____|OUT|
*  ___      R1 | ||  _|_  |     _|_   |___|    .SUBCKT  ONE_S  IN   OUT
* |IN |_/\ _|__ / \ |__ / \
* |___|  \/  IP| Egn1|   | Egn2|   ---->  |IN |_ _______ __|OUT|
*       10K   _| 90dB|  _|  -1 |   ---->  |___| |  1/s  |  |___|
*            |  \___/  |  \___/                 |_______|
*            |____|    |____|
*                /// ///                   s  = jw0     s = 2*PI*Freq
*                                          @s = jw0     R = 1/sC
*
*  http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html
*
*    ___         O1         C1 10pF    ___       .SUBCKT  OPA INP INN OUT
*   |INP|_      _____     _||______|OUT|
*   |___| |   _|_    |  R1 | ||  _|_  |___|           |\
*         |_ / \ |_/\ _|__ / \          |INP|__|-\
*           _| EGN1| 10K\/  IP| EGN2|  ---->  |___|  |  \__|OUT|
*   |INN|___|  -1 |       _| 90dB|  ---->           |  /  |___|
*   |___|    \___/   IP  |  \___/           |INN|__|+/
*              |          |____|            |___|  |/ OPA
*              |              |
*             ///            ///
.SUBCKT OPA  INP     INN     OUT
EGN1         O1      0       INP     INN     -1
EGN2         OUT     0       IP      0       -1000000
R1           O1      IP      10k
C1           OUT     IP      10p
.ends

.SUBCKT ONE_S IN       OUT
EGN1         O1       0      IP     0       -1000000
```

```
EGN2        OUT     0       O1      0       -1
R1          IN      IP      10k
C1          IP      O1      .01592u
.ends

.end

=====================END_OF_SPICE=======================
```

All resistor R1->R3 can be scaled up by 10K with no effect on frequency or Q.



It is the ratios of R1-> that control Q
And the integrator that controls the frequency.